

**CSAK A MEDVE EHET MÁLNÁT**


**CSEPPKÖVEK, JÉGKÉPZŐDMÉNYEK**

**A TÁTRA MADARAI**

**BÉKAKÖRKÉP**



TERMÉSZETTUDOMÁNYRÓL FIATALOKNAK / 1988.IV.ÉVF.



**10**

**OKTÓBER**

**ÁRA: 20,- Ft**



# DIGI SÜNI

## Számológató

Egy közismert játék számítógépes változatát közöljük most az Enterprise számítógép BASIC nyelvén. A játék lényege, hogy a táblázatban a számokat növekvő sorrendbe tologassuk. A joystickkel a „lyukat” tudjuk irányítani.

Ha sikerült, akkor START vagy FUNCTION 1 paranccsal indíthatod újra. Ha meguntad, és ismét a BASIC-et akarod használni, add ki előbb a CLEAR FONT parancsot.

Jó játékot!

FISCHER PÉTER

```

100 RANDOMIZE
110 CLEAR SCREEN
120 DIM TART(0 TO 15),TART0(0 TO 14),TART1(0 TO 14)
130 FOR N=0 TO 14
140 READ TART1(N)
150 NEXT
160 LET N=5+SPEEK(255,16374)+256*SPEEK(255,16375)
170 READ IF MISSING 200:M
180 LET N=N+1+SPOKE 255,N,M
190 GOTO 170
200 LET A$="" :LET X1=8:LET X2=15:LET TART(15)=16
210 SET CHARACTER 126,0,0,0,0,0,0,7,8,16
220 SET CHARACTER 125,0,0,0,0,0,0,255,0,0
230 SET CHARACTER 124,0,0,0,0,0,0,224,16,8
240 SET CHARACTER 123,16,17,18,20,20,20,20,20,20
250 SET CHARACTER 122,255,0,0,0,0,0,0,0,0
260 SET CHARACTER 121,128,65,34,20,20,20,20,20,20
270 SET CHARACTER 120,132,68,36,20,20,20,20,20,20
280 SET CHARACTER 119,20,20,20,20,20,20,20,20,20
290 SET CHARACTER 118,20,20,20,20,18,17,16,16,16
300 SET CHARACTER 117,20,20,20,20,34,65,128,0,0
310 SET CHARACTER 116,20,20,20,20,36,68,132,4,4
320 SET CHARACTER 115,15,0,0,0,0,0,0,0,0
330 SET CHARACTER 113,248,0,0,0,0,0,0,0,0
340 DATA 0,1,2,3,7,6,5,4,8,9,10,11,14,13,12
350 DATA 32,32,32,32,32,32,198,201,211,195,200,197,210,32,
208,197,212,197,210,32,32,177,185,184,184
360 PRINT CHR$(13),CHR$(13),CHR$(13),CHR$(13)
370 PRINT "      ~~~~~~"
380 FOR N=1 TO 4
390 PRINT "          (zzzyzyzyzzx"
400 PRINT "      w w w w w"
410 PRINT "      v))u))u))t"
420 NEXT N
430 PRINT "      szzzzzzzzzzq"
440 FOR N=0 TO 15
450 LET TART(N)=0
460 NEXT N
470 FOR N=1 TO 15
480 LET M=RND(15)
490 IF TART(M)>0 THEN GOTO 480
500 LET TART(M)=N:LET TART0(M)=N
510 NEXT N
520 LET I=0
530 LET HIBA=0
540 FOR N=1 TO 14
550 IF TART0(TART1(N))<TART0(TART1(N-1)) THEN LET M=TART0(TART1(N)):LET TART0(TART1(N))=TART0(TART1(N-1)):LET TART0(TART1(N-1))=M:LET I=I+1:LET HIBA=1
560 NEXT N
570 IF HIBA=1 THEN GOTO 530
580 IF MOD(I,2)=0 THEN GOTO 440
590 FOR N=0 TO 14
600 PRINT AT X1+3*INT(N/4),X2+3*MOD(N,4):;
610 PRINT USING 630:TART(N)
620 NEXT N
630 IMAGE:##

```



Monostori Csaba felvétele

```

640 LET N=15
650 LET HIBA=0
660 GET A$
670 IF A$="" THEN GOTO 660
680 SELECT CASE ORD(A$)
690 CASE 176
700 IF N<4 THEN
710 LET HIBA=1
720 ELSE
730 LET M=N-4
740 END IF
750 CASE 180
760 IF N>11 THEN
770 LET HIBA=1
780 ELSE
790 LET M=N+4
800 END IF
810 CASE 184
820 LET M=N-1
830 IF INT(M/4)<>INT(N/4) THEN LET HIBA=1
840 CASE 188
850 LET M=N+1
860 IF INT(M/4)<>INT(N/4) THEN LET HIBA=1
870 CASE ELSE
880 LET HIBA=1
890 END SELECT
900 IF HIBA=1 THEN GOTO 650
910 PRINT AT X1+3*INT(N/4),X2+3*MOD(N,4):;
920 PRINT " "
930 LET TART(N)=TART(M)
940 LET TART(M)=0
950 PRINT AT X1+3*INT(N/4),X2+3*MOD(N,4):;
960 PRINT USING 630:TART(N)
970 LET N=M
980 PRINT AT X1+3*INT(N/4),X2+3*MOD(N,4):;
990 PRINT " "
1000 IF N<15 THEN GOTO 660
1010 LET M=0
1020 DO WHILE TART(M)=M+1
1030 LET M=M+1
1040 LOOP
1050 IF M<15 THEN GOTO 660
1060 FOR N=0 TO 255
1070 SET #102:COLOR 1,N
1080 NEXT N
1090 WAIT 3
1100 TEXT

```

## CSEREBERE

COMMODORE-64 programokat cseréberélne Varga Péter. Címe: Letenye, Farkas J. u. 1. 4/2. 8868